feat: Create `code_vectors` table

lib/medical_transcription/code_vector.ex

@@ -0,0 +1,54 @@
+defmodule MedicalTranscription.CodeVector do
+  use Ecto.Schema
+  alias Ecto.Changeset
+  alias MedicalTranscription.Repo
+
+  import Ecto.Query
+  import Pgvector.Ecto.Query
+
+  schema "code_vectors" do
+    field :code, :string
+    field :description, :string
+    field :description_vector, Pgvector.Ecto.Vector
+  end
+
+  def changeset(code_vector, params \\ %{}) do
+    code_vector
+    |> Changeset.cast(params, [:code, :description, :description_vector])
+    |> Changeset.validate_required([:code, :description, :description_vector])
+  end
+
+  def insert_vector(params) do
+    changeset = changeset(%__MODULE__{}, params)
+
+    case Repo.insert(changeset) do
+      {:ok, _} ->
+        {:ok, "Success!"}
+
+      {:error, changeset} ->
+        messages =
+          Ecto.Changeset.traverse_errors(changeset, fn {msg, opts} ->
+            Enum.reduce(opts, msg, fn {key, value}, acc ->
+              String.replace(acc, "%{#{key}}", to_string(value))
+            end)
+          end)
+
+        {:error, messages}
+    end
+  end
+
+  def exists_for_code?(code) do
+    Repo.exists?(
+      from v in __MODULE__,
+        where: v.code == ^code
+    )
+  end
+
+  def find_similar(search_vector, limit \\ 5) do
+    Repo.all(
+      from v in __MODULE__,
+        order_by: cosine_distance(v.description_vector, ^search_vector),
+        limit: ^limit
+    )
+  end
+end

priv/repo/migrations/20240125151833_create_code_vectors.exs

@@ -0,0 +1,11 @@
+defmodule MedicalTranscription.Repo.Migrations.CreateLabelVectors do
+  use Ecto.Migration
+
+  def change do
+    create table("code_vectors") do
+      add :code, :string
+      add :description, :string
+      add :description_vector, :vector, size: 384
+    end
+  end
+end